

Farkas' Lemma and Motzkin's Transposition Theorem*

Ralph Bottesch Max W. Haslbeck René Thiemann

December 14, 2021

Abstract

We formalize a proof of Motzkin's transposition theorem and Farkas' lemma in Isabelle/HOL. Our proof is based on the formalization of the simplex algorithm which, given a set of linear constraints, either returns a satisfying assignment to the problem or detects unsatisfiability. By reusing facts about the simplex algorithm we show that a set of linear constraints is unsatisfiable if and only if there is a linear combination of the constraints which evaluates to a trivially unsatisfiable inequality.

Contents

1	Introduction	2
2	Farkas Coefficients via the Simplex Algorithm of Duterte and de Moura	3
2.1	Linear Inequalities	3
2.2	Farkas' Lemma on Layer 4	5
2.3	Farkas' Lemma on Layer 3	7
2.4	Farkas' Lemma on Layer 2	8
2.5	Farkas' Lemma on Layer 1	10
3	Corollaries from the Literature	12
3.1	Farkas' Lemma on Delta-Rationals	13
3.2	Motzkin's Transposition Theorem or the Kuhn-Fourier Theorem	13
3.3	Farkas' Lemma	13
3.4	Farkas Lemma for Matrices	14
4	Unsatisfiability over the Reals	16

*Supported by FWF (Austrian Science Fund) project Y757. The authors are listed in alphabetical order regardless of individual contributions or seniority.

1 Introduction

This formalization augments the existing formalization of the simplex algorithm [3, 5, 7]. Given a system of linear constraints, the simplex implementation in [3] produces either a satisfying assignment or a subset of the given constraints that is itself unsatisfiable. Here we prove some variants of Farkas' Lemma. In essence, it states that if a set of constraints is unsatisfiable, then there is a linear combination of these constraints that evaluates to an unsatisfiable inequality of the form $0 \leq c$, for some negative c .

Our proof of Farkas' Lemma [4, Cor. 7.1e] relies on the formalized simplex algorithm: Under the assumption that the algorithm has detected unsatisfiability, we show that there exist coefficients for the above-mentioned linear combination of the input constraints.

Since the formalized algorithm follows the structure of the simplex algorithm by Dutertre and de Moura [2], it first goes through a number of preprocessing phases, before starting the simplex procedure in earnest. These are relevant for proving Farkas' Lemma. We distinguish four *layers* of the algorithm; at each layer, it operates on data that is a refinement of the data available at the previous layer.

- *Layer 1. Data:* the input – a set of linear constraints with rational coefficients. These can be equalities or strict/non-strict inequalities. *Preprocessing:* Each equality is split into two non-strict inequalities, strict inequalities are replaced by non-strict inequalities involving δ -rationals.
- *Layer 2. Data:* a set of linear constraints that are non-strict inequalities with δ -rationals. *Preprocessing:* Linear constraints are simplified so that each constraint involves a single variable, by introducing so-called slack variables where necessary. The equations defining the slack variables are collected in a *tableau*. The constraints are normalized so that they are of the form $y \leq c$ or $y \geq c$ (these are called *atoms*).
- *Layer 3. Data:* A tableau and a set of atoms. Here the algorithm initializes the simplex algorithm.
- *Layer 4. Data:* A tableau, a set of atoms and an assignment of the variables. The simplex procedure is run.

At the point in the execution where the simplex algorithm detects unsatisfiability, we can directly obtain coefficients for the desired linear combination. However, these coefficients must then be propagated backwards through the different layers, where the constraints themselves have been modified, in order to obtain coefficients for a linear combination of *input* constraints. These propagation steps make up a large part of the formalized

proof, since we must show, at each of the layers 1–3, that the existence of coefficients at the layer below translates into the existence of such coefficients for the current layer. This means, in particular, that we formulate and prove a version of Farkas’ Lemma for each of the four layers, in terms of the data available at the respective level. The theorem we obtain at Layer 1 is actually a more general version of Farkas’ lemma, in the sense that it allows for strict as well as non-strict inequalities, known as Motzkin’s Transposition Theorem [4, Cor. 7.1k] or the Kuhn–Fourier Theorem [6, Thm. 1.1.9].

Since the implementation of the simplex algorithm in [3], which our work relies on, is restricted to systems of constraints over the rationals, this formalization is also subject to the same restriction.

2 Farkas Coefficients via the Simplex Algorithm of Duterte and de Moura

Let c_1, \dots, c_n be a finite list of linear inequalities. Let C be a list of pairs (r_i, c_i) where r_i is a rational number. We say that C is a list of *Farkas coefficients* if the sum of all products $r_i \cdot c_i$ results in an inequality that is trivially unsatisfiable.

Farkas’ Lemma states that a finite set of non-strict linear inequalities is unsatisfiable if and only if Farkas coefficients exist. We will prove this lemma with the help of the simplex algorithm of Duterte and de Moura’s.

Note that the simplex implementation works on four layers, and we will formulate and prove a variant of Farkas’ Lemma for each of these layers.

```
theory Farkas
  imports Simplex.Simplex
begin
```

2.1 Linear Inequalities

Both Farkas’ Lemma and Motzkin’s Transposition Theorem require linear combinations of inequalities. To this end we define one type that permits strict and non-strict inequalities which are always of the form “polynomial R constant” where R is either \leq or $<$. On this type we can then define a commutative monoid.

A type for the two relations: less-or-equal and less-than.

```
datatype le-rel = Leq-Rel | Lt-Rel
```

```
primrec rel-of :: le-rel  $\Rightarrow$  'a :: lrv  $\Rightarrow$  'a  $\Rightarrow$  bool where
  rel-of Leq-Rel = ( $\leq$ )
| rel-of Lt-Rel = ( $<$ )
```

```
instantiation le-rel :: comm-monoid-add begin
```

```
definition zero-le-rel = Leq-Rel
```

```

fun plus-le-rel where
  plus-le-rel Leq-Rel Leq-Rel = Leq-Rel
| plus-le-rel - - = Lt-Rel
instance
  <proof>
end

lemma Leq-Rel-0: Leq-Rel = 0 <proof>

datatype 'a le-constraint = Le-Constraint (lec-rel: le-rel) (lec-poly: linear-poly)
  (lec-const: 'a)

abbreviation (input) Leqc  $\equiv$  Le-Constraint Leq-Rel

instantiation le-constraint :: (lrv) comm-monoid-add begin
fun plus-le-constraint :: 'a le-constraint  $\Rightarrow$  'a le-constraint  $\Rightarrow$  'a le-constraint where
  plus-le-constraint (Le-Constraint r1 p1 c1) (Le-Constraint r2 p2 c2) =
    (Le-Constraint (r1 + r2) (p1 + p2) (c1 + c2))

definition zero-le-constraint :: 'a le-constraint where
  zero-le-constraint = Leqc 0 0

instance <proof>
end

primrec satisfiable-le-constraint :: 'a::lrv valuation  $\Rightarrow$  'a le-constraint  $\Rightarrow$  bool (infixl
 $\models_{le}$  100) where
  (v  $\models_{le}$  (Le-Constraint rel l r))  $\longleftrightarrow$  (rel-of rel (l\{v\}) r)

lemma satisfies-zero-le-constraint: v  $\models_{le}$  0
  <proof>

lemma satisfies-sum-le-constraints:
  assumes v  $\models_{le}$  c v  $\models_{le}$  d
  shows v  $\models_{le}$  (c + d)
  <proof>

lemma satisfies-sumlist-le-constraints:
  assumes  $\bigwedge$  c. c  $\in$  set (cs :: 'a :: lrv le-constraint list)  $\Longrightarrow$  v  $\models_{le}$  c
  shows v  $\models_{le}$  sum-list cs
  <proof>

lemma sum-list-lec:
  sum-list ls = Le-Constraint
    (sum-list (map lec-rel ls))
    (sum-list (map lec-poly ls))
    (sum-list (map lec-const ls))
  <proof>

```

lemma *sum-list-Leq-Rel*: $((\sum x \leftarrow C. \text{lec-rel } (f \ x)) = \text{Leq-Rel}) \longleftrightarrow (\forall x \in \text{set } C. \text{lec-rel } (f \ x) = \text{Leq-Rel})$
 <proof>

2.2 Farkas' Lemma on Layer 4

On layer 4 the algorithm works on a state containing a tableau, atoms (or bounds), an assignment and a satisfiability flag. Only non-strict inequalities appear at this level. In order to even state a variant of Farkas' Lemma on layer 4, we need conversions from atoms to non-strict constraints and then further to linear inequalities of type *le-constraint*. The latter conversion is a partial operation, since non-strict constraints of type *ns-constraint* permit greater-or-equal constraints, whereas *le-constraint* allows only less-or-equal.

The advantage of first going via *ns-constraint* is that this type permits a multiplication with arbitrary rational numbers (the direction of the inequality must be flipped when multiplying by a negative number, which is not possible with *le-constraint*).

instantiation *ns-constraint* :: (scaleRat) scaleRat

begin

fun *scaleRat-ns-constraint* :: rat \Rightarrow 'a *ns-constraint* \Rightarrow 'a *ns-constraint* **where**
scaleRat-ns-constraint r (LEQ-ns p c) =
 (if (r < 0) then GEQ-ns (r *R p) (r *R c) else LEQ-ns (r *R p) (r *R c))
 | *scaleRat-ns-constraint* r (GEQ-ns p c) =
 (if (r > 0) then GEQ-ns (r *R p) (r *R c) else LEQ-ns (r *R p) (r *R c))

instance <proof>

end

lemma *sat-scale-rat-ns*: **assumes** $v \models_{ns} ns$

shows $v \models_{ns} (f *R ns)$

<proof>

lemma *scaleRat-scaleRat-ns-constraint*: **assumes** $a \neq 0 \implies b \neq 0$

shows $a *R (b *R (c :: 'a :: \text{lrV } ns\text{-constraint})) = (a * b) *R c$

<proof>

fun *lec-of-ns* **where**

lec-of-ns (LEQ-ns p c) = (Leqc p c)

fun *is-leq-ns* **where**

is-leq-ns (LEQ-ns p c) = True

| *is-leq-ns* (GEQ-ns p c) = False

lemma *lec-of-ns*:

assumes *is-leq-ns* c

shows $(v \models_{le} \text{lec-of-ns } c) \longleftrightarrow (v \models_{ns} c)$

<proof>

fun *nsc-of-atom* **where**

nsc-of-atom (*Leq* *var* *b*) = *LEQ-ns* (*lp-monom* 1 *var*) *b*
| *nsc-of-atom* (*Geq* *var* *b*) = *GEQ-ns* (*lp-monom* 1 *var*) *b*

lemma *nsc-of-atom*: $v \models_{ns} \textit{nsc-of-atom } a \longleftrightarrow v \models_a a$
⟨*proof*⟩

We say that C is a list of Farkas coefficients for a given tableau t and atom set as , if it is a list of pairs (r, a) such that $a \in as$, r is non-zero, $r \cdot a$ is a ‘less-than-or-equal’-constraint, and the linear combination of inequalities must result in an inequality of the form $p \leq c$, where $c < 0$ and $t \models p = 0$.

definition *farkas-coefficients-atoms-tableau* **where**

farkas-coefficients-atoms-tableau ($as :: 'a :: lrv \textit{atom set}$) $t \ C = (\exists \ p \ c.$
 $(\forall (r, a) \in \textit{set } C. a \in as \wedge \textit{is-leq-ns } (r *R \textit{nsc-of-atom } a) \wedge r \neq 0) \wedge$
 $(\sum (r, a) \leftarrow C. \textit{lec-of-nsc } (r *R \textit{nsc-of-atom } a)) = \textit{Leqc } p \ c \wedge$
 $c < 0 \wedge$
 $(\forall \ v :: 'a \textit{ valuation}. v \models_t t \longrightarrow (p \{v\} = 0)))$

We first prove that if the check-function detects a conflict, then Farkas coefficients do exist for the tableau and atom set for which the conflict is detected.

definition *bound-atoms* :: $(i, 'a) \textit{state} \Rightarrow 'a \textit{atom set } (\mathcal{B}_A)$ **where**

bound-atoms $s = (\lambda(v, x). \textit{Geq } v \ x) \textit{ ' (set-of-map } (\mathcal{B}_l \ s)) \cup$
 $(\lambda(v, x). \textit{Leq } v \ x) \textit{ ' (set-of-map } (\mathcal{B}_u \ s))$

context *PivotUpdateMinVars*
begin

lemma *farkas-check*:

assumes *check*: $\textit{check } s' = s$ **and** $U: \mathcal{U} \ s \neg \mathcal{U} \ s'$
and *inv*: $\nabla \ s' \ \Delta \ (\mathcal{T} \ s') \models_{\textit{noIhs}} s' \ \diamond \ s'$
and *index*: $\textit{index-valid as } s'$
shows $\exists \ C. \textit{farkas-coefficients-atoms-tableau } (\textit{snd } ' \ as) \ (\mathcal{T} \ s') \ C$

⟨*proof*⟩

end

Next, we show that a conflict found by the assert-bound function also gives rise to Farkas coefficients.

context *Update*
begin

lemma *farkas-assert-bound*: **assumes** *inv*: $\neg \mathcal{U} \ s \models_{\textit{noIhs}} s \ \Delta \ (\mathcal{T} \ s) \ \nabla \ s \ \diamond \ s$
and *index*: $\textit{index-valid as } s$
and $U: \mathcal{U} \ (\textit{assert-bound ia } s)$

shows $\exists \ C. \textit{farkas-coefficients-atoms-tableau } (\textit{snd } ' \ (\textit{insert ia } as)) \ (\mathcal{T} \ s) \ C$
⟨*proof*⟩

end

Moreover, we prove that all other steps of the simplex algorithm on layer 4, such as pivoting, asserting bounds without conflict, etc., preserve Farkas coefficients.

lemma *farkas-coefficients-atoms-tableau-mono*: **assumes** $as \subseteq bs$
shows *farkas-coefficients-atoms-tableau* $as\ t\ C \implies \text{farkas-coefficients-atoms-tableau}\ bs\ t\ C$
<proof>

locale *AssertAllState'''* = *AssertAllState''* *init* *ass-bnd* *chk* + *Update* *update* +
PivotUpdateMinVars *eq-idx-for-lvar* *min-lvar-not-in-bounds* *min-rvar-incdec-eq*
pivot-and-update
for *init* **and** *ass-bnd* :: $'i \times 'a :: lrv\ atom \Rightarrow -$ **and** *chk* :: $('i, 'a)\ state \Rightarrow ('i, 'a)\ state$
and *update* :: $nat \Rightarrow 'a :: lrv \Rightarrow ('i, 'a)\ state \Rightarrow ('i, 'a)\ state$
and *eq-idx-for-lvar* :: $tableau \Rightarrow var \Rightarrow nat$ **and**
min-lvar-not-in-bounds :: $('i, 'a :: lrv)\ state \Rightarrow var\ option$ **and**
min-rvar-incdec-eq :: $('i, 'a)\ Direction \Rightarrow ('i, 'a)\ state \Rightarrow eq \Rightarrow 'i\ list + var$ **and**
pivot-and-update :: $var \Rightarrow var \Rightarrow 'a \Rightarrow ('i, 'a)\ state \Rightarrow ('i, 'a)\ state$
+ **assumes** *ass-bnd*: *ass-bnd* = *Update.assert-bound* *update* **and**
chk: *chk* = *PivotUpdateMinVars.check* *eq-idx-for-lvar* *min-lvar-not-in-bounds*
min-rvar-incdec-eq *pivot-and-update*

context *AssertAllState'''*
begin

lemma *farkas-assert-bound-loop*: **assumes** \mathcal{U} (*assert-bound-loop* *as* (*init* *t*))
and *norm*: $\Delta\ t$
shows $\exists\ C. \text{farkas-coefficients-atoms-tableau}\ (snd\ 'set\ as)\ t\ C$
<proof>

Now we get to the main result for layer 4: If the main algorithm returns unsat, then there are Farkas coefficients for the tableau and atom set that were given as input for this layer.

lemma *farkas-assert-all-state*: **assumes** $U: \mathcal{U}$ (*assert-all-state* *t* *as*)
and *norm*: $\Delta\ t$
shows $\exists\ C. \text{farkas-coefficients-atoms-tableau}\ (snd\ 'set\ as)\ t\ C$
<proof>

2.3 Farkas' Lemma on Layer 3

There is only a small difference between layers 3 and 4, namely that there is no simplex algorithm (*assert-all-state*) on layer 3, but just a tableau and atoms.

Hence, one task is to link the unsatisfiability flag on layer 4 with unsatisfiability of the original tableau and atoms (layer 3). This can be done via the existing soundness results of the simplex algorithm. Moreover, we give an easy proof that the existence of Farkas coefficients for a tableau and set of atoms implies unsatisfiability.

end

lemma *farkas-coefficients-atoms-tableau-unsat*:

assumes *farkas-coefficients-atoms-tableau as t C*

shows $\nexists v. v \models_t t \wedge v \models_{as} as$

<proof>

Next is the main result for layer 3: a tableau and a finite set of atoms are unsatisfiable if and only if there is a list of Farkas coefficients for the set of atoms and the tableau.

lemma *farkas-coefficients-atoms-tableau*: **assumes** *norm: Δt*

and *fin: finite as*

shows $(\exists C. \text{farkas-coefficients-atoms-tableau as t C}) \longleftrightarrow (\nexists v. v \models_t t \wedge v \models_{as} as)$

<proof>

2.4 Farkas' Lemma on Layer 2

The main difference between layers 2 and 3 is the introduction of slack-variables in layer 3 via the preprocess-function. Our task here is to show that Farkas coefficients at layer 3 (where slack-variables are used) can be converted into Farkas coefficients for layer 2 (before the preprocessing).

We also need to adapt the previous notion of Farkas coefficients, which was used in *farkas-coefficients-atoms-tableau*, for layer 2. At layer 3, Farkas coefficients are the coefficients in a linear combination of atoms that evaluates to an inequality of the form $p \leq c$, where p is a linear polynomial, $c < 0$, and $t \models p = 0$ holds. At layer 2, the atoms are replaced by non-strict constraints where the left-hand side is a polynomial in the original variables, but the corresponding linear combination (with Farkas coefficients) evaluates directly to the inequality $0 \leq c$, with $c < 0$. The implication $t \models p = 0$ is no longer possible in this layer, since there is no tableau t , nor is it needed, since p is 0. Thus, the statement defining Farkas coefficients must be changed accordingly.

definition *farkas-coefficients-ns where*

farkas-coefficients-ns ns C = $(\exists c.$

*$(\forall (r, n) \in \text{set } C. n \in ns \wedge \text{is-leq-ns } (r *R n) \wedge r \neq 0) \wedge$*

*$(\sum (r, n) \leftarrow C. \text{lec-of-ns } (r *R n)) = \text{Leq } 0 c \wedge$*

$c < 0$)

The easy part is to prove that Farkas coefficients imply unsatisfiability.

lemma *farkas-coefficients-ns-unsat*:

assumes *farkas-coefficients-ns ns C*

shows $\nexists v. v \models_{nss} ns$

<proof>

In order to eliminate the need for a tableau, we require the notion of an arbitrary substitution on polynomials, where all variables can be replaced at

once. The existing simplex formalization provides only a function to replace one variable at a time.

definition *subst-poly* :: (*var* \Rightarrow *linear-poly*) \Rightarrow *linear-poly* \Rightarrow *linear-poly* **where**
subst-poly σ $p = (\sum x \in \text{vars } p. \text{coeff } p \ x *R \ \sigma \ x)$

lemma *subst-poly-0[simp]*: *subst-poly* σ $0 = 0$ \langle *proof* \rangle

lemma *valuate-subst-poly*: (*subst-poly* σ p) $\llbracket v \rrbracket = (p \llbracket (\lambda x. ((\sigma \ x) \llbracket v \rrbracket)) \rrbracket)$
 \langle *proof* \rangle

lemma *subst-poly-add*: *subst-poly* σ $(p + q) = \text{subst-poly } \sigma \ p + \text{subst-poly } \sigma \ q$
 \langle *proof* \rangle

fun *subst-poly-lec* :: (*var* \Rightarrow *linear-poly*) \Rightarrow 'a *le-constraint* \Rightarrow 'a *le-constraint*
where

subst-poly-lec σ (*Le-Constraint rel* p c) = *Le-Constraint rel* (*subst-poly* σ p) c

lemma *subst-poly-lec-0[simp]*: *subst-poly-lec* σ $0 = 0$ \langle *proof* \rangle

lemma *subst-poly-lec-add*: *subst-poly-lec* σ $(c1 + c2) = \text{subst-poly-lec } \sigma \ c1 + \text{subst-poly-lec } \sigma \ c2$
 \langle *proof* \rangle

lemma *subst-poly-lec-sum-list*: *subst-poly-lec* σ (*sum-list* ps) = *sum-list* (*map* (*subst-poly-lec* σ) ps)
 \langle *proof* \rangle

lemma *subst-poly-lp-monom[simp]*: *subst-poly* σ (*lp-monom* r x) = $r *R \ \sigma \ x$
 \langle *proof* \rangle

lemma *subst-poly-scaleRat*: *subst-poly* σ $(r *R \ p) = r *R (\text{subst-poly } \sigma \ p)$
 \langle *proof* \rangle

We need several auxiliary properties of the preprocess-function which are not present in the simplex formalization.

lemma *Tableau-is-monom-preprocess'*:
assumes $(x, p) \in \text{set } (\text{Tableau } (\text{preprocess}' \ cs \ \text{start}))$
shows $\neg \text{is-monom } p$
 \langle *proof* \rangle

lemma *preprocess'-atoms-to-constraints'*: **assumes** *preprocess'* cs $\text{start} = S$
shows $\text{set } (\text{Atoms } S) \subseteq \{(i, \text{qdelta-constraint-to-atom } c \ v) \mid i \ c \ v. (i, c) \in \text{set } cs\}$
 \wedge
 $(\neg \text{is-monom } (\text{poly } c) \longrightarrow \text{Poly-Mapping } S (\text{poly } c) = \text{Some } v)$
 \langle *proof* \rangle

lemma *monom-of-atom-coeff*:
assumes *is-monom* $(\text{poly } ns)$ $a = \text{qdelta-constraint-to-atom } ns \ v$
shows $(\text{monom-coeff } (\text{poly } ns)) *R \ \text{nsc-of-atom } a = ns$

<proof>

The next lemma provides the functionality that is required to convert an atom back to a non-strict constraint, i.e., it is a kind of inverse of the preprocess-function.

lemma *preprocess'-atoms-to-constraints*: **assumes** S : *preprocess' cs start = S*
and *start*: *start = start-fresh-variable cs*
and *ns*: $ns = (\text{case } a \text{ of } \text{Leq } v \ c \Rightarrow \text{LEQ-ns } q \ c \mid \text{Geq } v \ c \Rightarrow \text{GEQ-ns } q \ c)$
and $a \in \text{snd ' set (Atoms } S)$
shows $(\text{atom-var } a \notin \text{fst ' set (Tableau } S) \longrightarrow (\exists r. r \neq 0 \wedge r * R \text{ nsc-of-atom } a \in \text{snd ' set } cs))$
 $\wedge ((\text{atom-var } a, q) \in \text{set (Tableau } S) \longrightarrow ns \in \text{snd ' set } cs)$
<proof>

Next follows the major technical lemma of this part, namely that Farkas coefficients on layer 3 for preprocessed constraints can be converted into Farkas coefficients on layer 2.

lemma *farkas-coefficients-preprocess'*:
assumes *pp*: *preprocess' cs (start-fresh-variable cs) = S* **and**
ft: *farkas-coefficients-atoms-tableau (snd ' set (Atoms } S)) (Tableau } S) C*
shows $\exists C. \text{farkas-coefficients-ns (snd ' set } cs) C$
<proof>

lemma *preprocess'-unsat-indexD*: $i \in \text{set (UnsatIndices (preprocess' ns } j)) \implies$
 $\exists c. \text{poly } c = 0 \wedge \neg \text{zero-satisfies } c \wedge (i, c) \in \text{set } ns$
<proof>

lemma *preprocess'-unsat-index-farkas-coefficients-ns*:
assumes $i \in \text{set (UnsatIndices (preprocess' ns } j))$
shows $\exists C. \text{farkas-coefficients-ns (snd ' set } ns) C$
<proof>

The combination of the previous results easily provides the main result of this section: a finite set of non-strict constraints on layer 2 is unsatisfiable if and only if there are Farkas coefficients. Again, here we use results from the simplex formalization, namely soundness of the preprocess-function.

lemma *farkas-coefficients-ns*: **assumes** *finite* ($ns :: Q\Delta$ *ns-constraint set*)
shows $(\exists C. \text{farkas-coefficients-ns } ns \ C) \longleftrightarrow (\nexists v. v \models_{nss} ns)$
<proof>

2.5 Farkas' Lemma on Layer 1

The main difference of layers 1 and 2 is the restriction to non-strict constraints via delta-rationals. Since we now work with another constraint type, *constraint*, we again need translations into linear inequalities of type *le-constraint*. Moreover, we also need to define scaling of constraints where flipping the comparison sign may be required.

fun *is-le* :: *constraint* \Rightarrow *bool* **where**

```

  is-le (LT - -) = True
| is-le (LEQ - -) = True
| is-le - = False

```

fun *lec-of-constraint* **where**

```

  lec-of-constraint (LEQ p c) = (Le-Constraint Leq-Rel p c)
| lec-of-constraint (LT p c) = (Le-Constraint Lt-Rel p c)

```

lemma *lec-of-constraint*:

```

assumes is-le c
shows (v  $\models_{le}$  (lec-of-constraint c))  $\longleftrightarrow$  (v  $\models_c$  c)
<proof>

```

instantiation *constraint* :: *scaleRat*

begin

fun *scaleRat-constraint* :: *rat* \Rightarrow *constraint* \Rightarrow *constraint* **where**

```

  scaleRat-constraint r cc = (if r = 0 then LEQ 0 0 else
    (case cc of
      LEQ p c  $\Rightarrow$ 
        (if (r < 0) then GEQ (r *R p) (r *R c) else LEQ (r *R p) (r *R c))
    | LT p c  $\Rightarrow$ 
        (if (r < 0) then GT (r *R p) (r *R c) else LT (r *R p) (r *R c))
    | GEQ p c  $\Rightarrow$ 
        (if (r > 0) then GEQ (r *R p) (r *R c) else LEQ (r *R p) (r *R c))
    | GT p c  $\Rightarrow$ 
        (if (r > 0) then GT (r *R p) (r *R c) else LT (r *R p) (r *R c))
    | LTPP p q  $\Rightarrow$ 
        (if (r < 0) then GT (r *R (p - q)) 0 else LT (r *R (p - q)) 0)
    | LEQPP p q  $\Rightarrow$ 
        (if (r < 0) then GEQ (r *R (p - q)) 0 else LEQ (r *R (p - q)) 0)
    | GTPP p q  $\Rightarrow$ 
        (if (r > 0) then GT (r *R (p - q)) 0 else LT (r *R (p - q)) 0)
    | GEQPP p q  $\Rightarrow$ 
        (if (r > 0) then GEQ (r *R (p - q)) 0 else LEQ (r *R (p - q)) 0)
    | EQPP p q  $\Rightarrow$  LEQ (r *R (p - q)) 0 — We do not keep equality, since the aim
    is to convert the scaled constraints into inequalities, which will then be summed
    up.
    | EQ p c  $\Rightarrow$  LEQ (r *R p) (r *R c)
  ))

```

instance <proof>

end

lemma *sat-scale-rat*: **assumes** (v :: *rat valuation*) \models_c c

```

shows v  $\models_c$  (r *R c)
<proof>

```

In the following definition of Farkas coefficients (for layer 1), the main difference to *farkas-coefficients-ns* is that the linear combination evaluates either to a strict inequality where the constant must be non-positive, or to

a non-strict inequality where the constant must be negative.

definition *farkas-coefficients where*

farkas-coefficients cs C = $(\exists d \text{ rel. } (\forall (r,c) \in \text{set } C. c \in \text{cs} \wedge \text{is-le } (r *R c) \wedge r \neq 0) \wedge (\sum (r,c) \leftarrow C. \text{lec-of-constraint } (r *R c)) = \text{Le-Constraint rel } 0 d \wedge (\text{rel} = \text{Leq-Rel} \wedge d < 0 \vee \text{rel} = \text{Lt-Rel} \wedge d \leq 0))$

Again, the existence Farkas coefficients immediately implies unsatisfiability.

lemma *farkas-coefficients-unsat:*

assumes *farkas-coefficients cs C*

shows $\nexists v. v \models_{cs} cs$

<proof>

Now follows the difficult implication. The major part is proving that the translation *constraint-to-qdelta-constraint* preserves the existence of Farkas coefficients via pointwise compatibility of the sum. Here, compatibility links a strict or non-strict inequality from the input constraint to a translated non-strict inequality over delta-rationals.

fun *compatible-cs where*

compatible-cs (Le-Constraint Leq-Rel p c) (Le-Constraint Leq-Rel q d) = $(q = p \wedge d = QDelta c 0)$

| compatible-cs (Le-Constraint Lt-Rel p c) (Le-Constraint Leq-Rel q d) = $(q = p \wedge qdfst d = c)$

| compatible-cs - - = False

lemma *compatible-cs-0-0: compatible-cs 0 0 <proof>*

lemma *compatible-cs-plus: compatible-cs c1 d1 \implies compatible-cs c2 d2 \implies compatible-cs (c1 + c2) (d1 + d2)*

<proof>

lemma *unsat-farkas-coefficients: assumes* $\nexists v. v \models_{cs} cs$

and *fin: finite cs*

shows $\exists C. \text{farkas-coefficients cs } C$

<proof>

Finally we can prove on layer 1 that a finite set of constraints is unsatisfiable if and only if there are Farkas coefficients.

lemma *farkas-coefficients: assumes finite cs*

shows $(\exists C. \text{farkas-coefficients cs } C) \longleftrightarrow (\nexists v. v \models_{cs} cs)$

<proof>

3 Corollaries from the Literature

In this section, we convert the previous variations of Farkas' Lemma into more well-known forms of this result. Moreover, instead of referring to the

various constraint types of the simplex formalization, we now speak solely about constraints of type *le-constraint*.

3.1 Farkas' Lemma on Delta-Rationals

We start with Lemma 2 of [1], a variant of Farkas' Lemma for delta-rationals. To be more precise, it states that a set of non-strict inequalities over delta-rationals is unsatisfiable if and only if there is a linear combination of the inequalities that results in a trivial unsatisfiable constraint $0 < \text{const}$ for some negative constant *const*. We can easily prove this statement via the lemma *farkas-coefficients-ns* and some conversions between the different constraint types.

lemma *Farkas'-Lemma-Delta-Rationals*: **fixes** *cs* :: *QDelta le-constraint set*
assumes *only-non-strict*: *lec-rel* ' *cs* \subseteq {*Leq-Rel*}
and *fin*: *finite cs*
shows ($\nexists v. \forall c \in cs. v \models_{le} c$) \longleftrightarrow
 $(\exists C \text{ const}. (\forall (r, c) \in \text{set } C. r > 0 \wedge c \in cs)$
 $\wedge (\sum (r, c) \leftarrow C. \text{Leqc } (r *R \text{lec-poly } c) (r *R \text{lec-const } c)) = \text{Leqc } 0 \text{ const}$
 $\wedge \text{const} < 0)$
(is *?lhs = ?rhs*)
<proof>

3.2 Motzkin's Transposition Theorem or the Kuhn-Fourier Theorem

Next, we prove a generalization of Farkas' Lemma that permits arbitrary combinations of strict and non-strict inequalities: Motzkin's Transposition Theorem which is also known as the Kuhn-Fourier Theorem.

The proof is mainly based on the lemma *farkas-coefficients*, again requiring conversions between constraint types.

theorem *Motzkin's-transposition-theorem*: **fixes** *cs* :: *rat le-constraint set*
assumes *fin*: *finite cs*
shows ($\nexists v. \forall c \in cs. v \models_{le} c$) \longleftrightarrow
 $(\exists C \text{ const rel}. (\forall (r, c) \in \text{set } C. r > 0 \wedge c \in cs)$
 $\wedge (\sum (r, c) \leftarrow C. \text{Le-Constraint } (\text{lec-rel } c) (r *R \text{lec-poly } c) (r *R \text{lec-const}$
 $c))$
 $= \text{Le-Constraint rel } 0 \text{ const}$
 $\wedge (\text{rel} = \text{Leq-Rel} \wedge \text{const} < 0 \vee \text{rel} = \text{Lt-Rel} \wedge \text{const} \leq 0))$
(is *?lhs = ?rhs*)
<proof>

3.3 Farkas' Lemma

Finally we derive the commonly used form of Farkas' Lemma, which easily follows from *Motzkin's-transposition-theorem*. It only permits non-strict inequalities and, as a result, the sum of inequalities will always be non-strict.

lemma *Farkas'-Lemma*: **fixes** $cs :: \text{rat le-constraint set}$
assumes *only-non-strict*: $\text{lec-rel } 'cs \subseteq \{\text{Leq-Rel}\}$
and *fin*: *finite cs*
shows $(\nexists v. \forall c \in cs. v \models_{le} c) \longleftrightarrow$
 $(\exists C \text{ const. } (\forall (r, c) \in \text{set } C. r > 0 \wedge c \in cs)$
 $\wedge (\sum (r, c) \leftarrow C. \text{Leqc } (r *R \text{lec-poly } c) (r *R \text{lec-const } c)) = \text{Leqc } 0 \text{ const}$
 $\wedge \text{const} < 0)$
(is - = ?rhs)
 $\langle \text{proof} \rangle$

We also present slightly modified versions

lemma *sum-list-map-filter-sum*: **fixes** $f :: 'a \Rightarrow 'b :: \text{comm-monoid-add}$
shows $\text{sum-list } (\text{map } f (\text{filter } g \text{ xs})) + \text{sum-list } (\text{map } f (\text{filter } (\text{Not } o \text{ } g) \text{ xs})) =$
 $\text{sum-list } (\text{map } f \text{ xs})$
 $\langle \text{proof} \rangle$

A version where every constraint obtains exactly one coefficient and where 0 coefficients are allowed.

lemma *Farkas'-Lemma-set-sum*: **fixes** $cs :: \text{rat le-constraint set}$
assumes *only-non-strict*: $\text{lec-rel } 'cs \subseteq \{\text{Leq-Rel}\}$
and *fin*: *finite cs*
shows $(\nexists v. \forall c \in cs. v \models_{le} c) \longleftrightarrow$
 $(\exists C \text{ const. } (\forall c \in cs. C \ c \geq 0)$
 $\wedge (\sum c \in cs. \text{Leqc } ((C \ c) *R \text{lec-poly } c) ((C \ c) *R \text{lec-const } c)) = \text{Leqc } 0$
 const
 $\wedge \text{const} < 0)$
 $\langle \text{proof} \rangle$

A version with indexed constraints, i.e., in particular where constraints may occur several times.

lemma *Farkas'-Lemma-indexed*: **fixes** $c :: \text{nat} \Rightarrow \text{rat le-constraint}$
assumes *only-non-strict*: $\text{lec-rel } 'c \ 'Is \subseteq \{\text{Leq-Rel}\}$
and *fin*: *finite Is*
shows $(\nexists v. \forall i \in Is. v \models_{le} c \ i) \longleftrightarrow$
 $(\exists C \text{ const. } (\forall i \in Is. C \ i \geq 0)$
 $\wedge (\sum i \in Is. \text{Leqc } ((C \ i) *R \text{lec-poly } (c \ i)) ((C \ i) *R \text{lec-const } (c \ i))) =$
 $\text{Leqc } 0 \text{ const}$
 $\wedge \text{const} < 0)$
 $\langle \text{proof} \rangle$

end

3.4 Farkas Lemma for Matrices

In this part we convert the simplex-structures like linear polynomials, etc., into equivalent formulations using matrices and vectors. As a result we present Farkas' Lemma via matrices and vectors.

```

theory Matrix-Farkas
  imports Farkas
    Jordan-Normal-Form.Matrix
begin

lift-definition poly-of-vec :: rat vec  $\Rightarrow$  linear-poly is
   $\lambda v x.$  if  $(x < \text{dim-vec } v)$  then  $v \$ x$  else 0
   $\langle \text{proof} \rangle$ 

definition val-of-vec :: rat vec  $\Rightarrow$  rat valuation where
  val-of-vec  $v x = v \$ x$ 

lemma valuate-poly-of-vec: assumes  $w \in \text{carrier-vec } n$ 
  and  $v \in \text{carrier-vec } n$ 
shows valuate (poly-of-vec  $v$ ) (val-of-vec  $w$ ) =  $v \cdot w$ 
   $\langle \text{proof} \rangle$ 

definition constraints-of-mat-vec :: rat mat  $\Rightarrow$  rat vec  $\Rightarrow$  rat le-constraint set
where
  constraints-of-mat-vec  $A b = (\lambda i . \text{Leqc } (\text{poly-of-vec } (\text{row } A \ i)) (b \$ i)) \text{ ' } \{0 .. < \text{dim-row } A\}$ 

lemma constraints-of-mat-vec-solution-main: assumes  $A: A \in \text{carrier-mat } nr \ nc$ 
  and  $x: x \in \text{carrier-vec } nc$ 
  and  $b: b \in \text{carrier-vec } nr$ 
  and  $\text{sol}: A *_v x \leq b$ 
  and  $c: c \in \text{constraints-of-mat-vec } A \ b$ 
shows val-of-vec  $x \models_{le} c$ 
   $\langle \text{proof} \rangle$ 

lemma vars-poly-of-vec: vars (poly-of-vec  $v$ )  $\subseteq \{0 .. < \text{dim-vec } v\}$ 
   $\langle \text{proof} \rangle$ 

lemma finite-constraints-of-mat-vec: finite (constraints-of-mat-vec  $A \ b$ )
   $\langle \text{proof} \rangle$ 

lemma lec-rec-constraints-of-mat-vec: lec-rel ' constraints-of-mat-vec  $A \ b \subseteq \{\text{Leq-Rel}\}$ 
   $\langle \text{proof} \rangle$ 

lemma constraints-of-mat-vec-solution-1:
  assumes  $A: A \in \text{carrier-mat } nr \ nc$ 
  and  $b: b \in \text{carrier-vec } nr$ 
  and  $\text{sol}: \exists x \in \text{carrier-vec } nc. A *_v x \leq b$ 
shows  $\exists v. \forall c \in \text{constraints-of-mat-vec } A \ b. v \models_{le} c$ 
   $\langle \text{proof} \rangle$ 

lemma constraints-of-mat-vec-solution-2:
  assumes  $A: A \in \text{carrier-mat } nr \ nc$ 

```

and $b: b \in \text{carrier-vec } nr$
and $\text{sol}: \exists v. \forall c \in \text{constraints-of-mat-vec } A \ b. v \models_{le} c$
shows $\exists x \in \text{carrier-vec } nc. A *_v x \leq b$
 $\langle \text{proof} \rangle$

lemma *constraints-of-mat-vec-solution*:
assumes $A: A \in \text{carrier-mat } nr \ nc$
and $b: b \in \text{carrier-vec } nr$
shows $(\exists x \in \text{carrier-vec } nc. A *_v x \leq b) =$
 $(\exists v. \forall c \in \text{constraints-of-mat-vec } A \ b. v \models_{le} c)$
 $\langle \text{proof} \rangle$

lemma *farkas-lemma-matrix*: **fixes** $A :: \text{rat mat}$
assumes $A: A \in \text{carrier-mat } nr \ nc$
and $b: b \in \text{carrier-vec } nr$
shows $(\exists x \in \text{carrier-vec } nc. A *_v x \leq b) \longleftrightarrow$
 $(\forall y. y \geq 0_v \ nr \longrightarrow \text{mat-of-row } y * A = 0_m \ 1 \ nc \longrightarrow y \cdot b \geq 0)$
 $\langle \text{proof} \rangle$

lemma *farkas-lemma-matrix'*: **fixes** $A :: \text{rat mat}$
assumes $A: A \in \text{carrier-mat } nr \ nc$
and $b: b \in \text{carrier-vec } nr$
shows $(\exists x \geq 0_v \ nc. A *_v x = b) \longleftrightarrow$
 $(\forall y \in \text{carrier-vec } nr. \text{mat-of-row } y * A \geq 0_m \ 1 \ nc \longrightarrow y \cdot b \geq 0)$
 $\langle \text{proof} \rangle$

end

4 Unsatisfiability over the Reals

By using Farkas' Lemma we prove that a finite set of linear rational inequalities is satisfiable over the rational numbers if and only if it is satisfiable over the real numbers. Hence, the simplex algorithm either gives a rational solution or shows unsatisfiability over the real numbers.

theory *Simplex-for-Reals*
imports
 $Farkas$
 $Simplex.Simplex-Incremental$
begin

instantiation $real :: lrv$
begin
definition $\text{scaleRat-real} :: \text{rat} \Rightarrow \text{real} \Rightarrow \text{real}$ **where**
 $[\text{simp}]: x *R y = \text{real-of-rat } x * y$
instance $\langle \text{proof} \rangle$
end

abbreviation *real-satisfies-constraints* :: *real valuation* \Rightarrow *constraint set* \Rightarrow *bool*
(infixl \models_{rcs} *100*) **where**

$$v \models_{rcs} cs \equiv \forall c \in cs. v \models_c c$$

definition *of-rat-val* :: *rat valuation* \Rightarrow *real valuation* **where**
of-rat-val $v\ x = \text{of-rat}\ (v\ x)$

lemma *of-rat-val-eval*: $p\ \{\!\{ \text{of-rat-val}\ v\ \}\!\} = \text{of-rat}\ (p\ \{\!\{v\}\!\})$
 $\langle \text{proof} \rangle$

lemma *of-rat-val-constraint*: $\text{of-rat-val}\ v \models_c c \longleftrightarrow v \models_c c$
 $\langle \text{proof} \rangle$

lemma *of-rat-val-constraints*: $\text{of-rat-val}\ v \models_{rcs} cs \longleftrightarrow v \models_{cs} cs$
 $\langle \text{proof} \rangle$

lemma *sat-scale-rat-real*: **assumes** $(v :: \text{real valuation}) \models_c c$
shows $v \models_c (r * R\ c)$
 $\langle \text{proof} \rangle$

fun *of-rat-lec* :: *rat le-constraint* \Rightarrow *real le-constraint* **where**
of-rat-lec $(\text{Le-Constraint}\ r\ p\ c) = \text{Le-Constraint}\ r\ p\ (\text{of-rat}\ c)$

lemma *lec-of-constraint-real*:
assumes *is-le* c
shows $(v \models_{le} \text{of-rat-lec}\ (\text{lec-of-constraint}\ c)) \longleftrightarrow (v \models_c c)$
 $\langle \text{proof} \rangle$

lemma *of-rat-lec-add*: $\text{of-rat-lec}\ (c + d) = \text{of-rat-lec}\ c + \text{of-rat-lec}\ d$
 $\langle \text{proof} \rangle$

lemma *of-rat-lec-zero*: $\text{of-rat-lec}\ 0 = 0$
 $\langle \text{proof} \rangle$

lemma *of-rat-lec-sum*: $\text{of-rat-lec}\ (\text{sum-list}\ c) = \text{sum-list}\ (\text{map}\ \text{of-rat-lec}\ c)$
 $\langle \text{proof} \rangle$

This is the main lemma: a finite set of linear constraints is satisfiable over \mathbb{Q} if and only if it is satisfiable over \mathbb{R} .

lemma *rat-real-conversion*: **assumes** *finite* cs
shows $(\exists v :: \text{rat valuation}. v \models_{cs} cs) \longleftrightarrow (\exists v :: \text{real valuation}. v \models_{rcs} cs)$
 $\langle \text{proof} \rangle$

The main result of simplex, now using unsatisfiability over the reals.

fun *i-satisfies-cs-real* **(infixl** \models_{rics} *100*) **where**
 $(I, v) \models_{rics} cs \longleftrightarrow v \models_{rcs} \text{Simplex.restrict-to}\ I\ cs$

lemma *simplex-index-real*:
 $\text{simplex-index}\ cs = \text{Unsat}\ I \implies \text{set}\ I \subseteq \text{fst}\ ' \text{set}\ cs \wedge \neg (\exists v. (\text{set}\ I, v) \models_{rics} \text{set}\ cs) \wedge$

(*distinct-indices cs* \longrightarrow ($\forall J \subset \text{set } I. (\exists v. (J, v) \models_{ics} \text{set } cs)$)) — minimal unsat core over the reals
simplex-index cs = *Sat v* $\implies \langle v \rangle \models_{cs} (\text{snd } ' \text{set } cs)$ — satisfying assignment
 $\langle \text{proof} \rangle$

lemma *simplex-real*:

simplex cs = *Unsat I* $\implies \neg (\exists v. v \models_{rcs} \text{set } cs)$ — unsat of original constraints over the reals

simplex cs = *Unsat I* $\implies \text{set } I \subseteq \{0..<\text{length } cs\} \wedge \neg (\exists v. v \models_{rcs} \{cs ! i \mid i. i \in \text{set } I\})$

$\wedge (\forall J \subset \text{set } I. \exists v. v \models_{cs} \{cs ! i \mid i. i \in J\})$ — minimal unsat core over reals

simplex cs = *Sat v* $\implies \langle v \rangle \models_{cs} \text{set } cs$ — satisfying assignment over the rationals
 $\langle \text{proof} \rangle$

Define notion of minimal unsat core over the reals: the subset has to be unsat over the reals, and every proper subset has to be satisfiable over the rational numbers.

definition *minimal-unsat-core-real* :: 'i set \Rightarrow 'i i-constraint list \Rightarrow bool **where**

minimal-unsat-core-real I ics = (($I \subseteq \text{fst } ' \text{set } ics$) \wedge ($\neg (\exists v. (I, v) \models_{rics} \text{set } ics)$))

\wedge (*distinct-indices ics* \longrightarrow ($\forall J. J \subset I \longrightarrow (\exists v. (J, v) \models_{ics} \text{set } ics)$)))

Because of equi-satisfiability the two notions of minimal unsat cores coincide.

lemma *minimal-unsat-core-real-conv*: *minimal-unsat-core-real I ics* = *minimal-unsat-core I ics*

$\langle \text{proof} \rangle$

Easy consequence: The incremental simplex algorithm is also sound wrt. minimal-unsat-cores over the reals.

lemmas *incremental-simplex-real* =

init-simplex

assert-simplex-ok

assert-simplex-unsat[*folded minimal-unsat-core-real-conv*]

assert-all-simplex-ok

assert-all-simplex-unsat[*folded minimal-unsat-core-real-conv*]

check-simplex-ok

check-simplex-unsat[*folded minimal-unsat-core-real-conv*]

solution-simplex

backtrack-simplex

checked-invariant-simplex

end

References

- [1] M. Bromberger and C. Weidenbach. New techniques for linear arithmetic: cubes and equalities. *Formal Methods in System Design*, 51(3):433–461, Dec 2017.
- [2] B. Dutertre and L. M. de Moura. A fast linear-arithmetic solver for DPLL(T). In T. Ball and R. B. Jones, editors, *CAV'06*, volume 4144 of *LNCS*, pages 81–94, 2006.
- [3] F. Marić, M. Spasić, and R. Thiemann. An incremental simplex algorithm with unsatisfiable core generation. *Archive of Formal Proofs*, Aug. 2018. <http://isa-afp.org/entries/Simplex.html>, Formal proof development.
- [4] A. Schrijver. *Theory of linear and integer programming*. Wiley, 1999.
- [5] M. Spasić and F. Marić. Formalization of incremental simplex algorithm by stepwise refinement. In D. Giannakopoulou and D. Méry, editors, *FM'12*, volume 7436 of *LNCS*, pages 434–449, 2012.
- [6] J. Stoer and C. Witzgall. *Convexity and Optimization in Finite Dimensions I*. Die Grundlehren der mathematischen Wissenschaften 163. Springer-Verlag Berlin Heidelberg, 1 edition, 1970.
- [7] R. Thiemann. Extending a verified simplex algorithm. In G. Barthe, K. Korovin, S. Schulz, M. Suda, G. Sutcliffe, and M. Veanes, editors, *LPAR-22 Workshop and Short Paper Proceedings*, volume 9 of *Kalpa Publications in Computing*, pages 37–48. EasyChair, 2018.